These examples are from the handout on how to use power series in Maple.

Example 1: Solve y''+xy'+2y=0, y(0)=a_0 y'(0)=a_1

```
> with(powseries):
> deq:=diff(y(x), x$2) + x*diff(y(x), x) + 2*y(x) = 0;
> inits:=y(0)=a[0],D(y)(0)=a[1];
> IVP:={deq,inits};
```

$$deq := \frac{d^2}{dx^2}\, y(x) + x \left(\frac{d}{dx}\, y(x)\right) + 2\, y(x) = 0$$

$$inits := y(0) = a_0,\ D(y)(0) = a_1$$

$$IVP := \left\{\frac{d^2}{dx^2}\, y(x) + x \left(\frac{d}{dx}\, y(x)\right) + 2\, y(x) = 0,\ y(0) = a_0,\ D(y)(0) = a_1\right\} \quad (1)$$

```
> #The following computes the series solution
> #   f is a procedure, F is the series

> f:=powsolve(IVP);
> F:=tpsform(f,x,12);
```

$$f := \mathbf{proc}(powparm)\ \ldots\ \mathbf{end\ proc}$$

$$F := a_0 + a_1\, x - a_0\, x^2 - \frac{1}{2}\, a_1\, x^3 + \frac{1}{3}\, a_0\, x^4 + \frac{1}{8}\, a_1\, x^5 - \frac{1}{15}\, a_0\, x^6 - \frac{1}{48}\, a_1\, x^7 + \frac{1}{105}\, a_0\, x^8 + \frac{1}{384}\, a_1\, x^9 - \frac{1}{945}\, a_0\, x^{10} - \frac{1}{3840}\, a_1\, x^{11} + O(x^{12}) \quad (2)$$

```
> #We can extract the recursion:
> f(_k);
```

$$-\frac{a(\_k - 2)}{\_k - 1} \quad (3)$$

```
> #or, more succinctly:
> recursion_relation:=a(n)=subs(_k=n,f(_k));
```

$$recursion\_relation := a(n) = -\frac{a(n-2)}{n-1} \quad (4)$$

Example 2: Here, we'll also graph the solution. First, clear the function from the last example, then start things up.

```
> unassign('y');

> eqn:=diff(y(x),x$2)-(3*x-2)*diff(y(x),x)-2*y(x)=0;
> inits:=y(0)=0,D(y)(0)=2;
> IVP:={eqn,inits};
```

$$eqn := \frac{d^2}{dx^2}\, y(x) - (3\,x-2)\left(\frac{d}{dx}\, y(x)\right) - 2\,y(x) = 0$$

$$inits := y(0) = 0, D(y)(0) = 2$$

$$IVP := \left\{\frac{d^2}{dx^2}\, y(x) - (3\,x-2)\left(\frac{d}{dx}\, y(x)\right) - 2\,y(x) = 0, y(0) = 0, D(y)(0) = 2\right\} \quad \textbf{(5)}$$

```
> with(powseries):
> f:=powsolve(IVP);
> recursion_relation:=a(n)=subs(_k=n,f(_k));
```

$$f := \textbf{proc}(powparm) \;\ldots\; \textbf{end proc}$$

$$recursion\_relation := a(n) = \frac{-2\,a(n-1)\,n + 2\,a(n-1) - 4\,a(n-2) + 3\,a(n-2)\,n}{n\,(n-1)} \quad \textbf{(6)}$$

```
> f6:=tpsform(f,x,7);
> f12:=tpsform(f,x,13);
```

$$f6 := 2\,x - 2\,x^2 + 3\,x^3 - \frac{17}{6}\,x^4 + \frac{167}{60}\,x^5 - \frac{9}{4}\,x^6 + O(x^7)$$

$$f12 := 2\,x - 2\,x^2 + 3\,x^3 - \frac{17}{6}\,x^4 + \frac{167}{60}\,x^5 - \frac{9}{4}\,x^6 + \frac{637}{360}\,x^7 - \frac{12559}{10080}\,x^8 + \frac{5659}{6720}\,x^9 - \frac{479327}{907200}\,x^{10} + \frac{1269661}{3991680}\,x^{11} - \frac{2409641}{13305600}\,x^{12} + O(x^{13}) \quad \textbf{(7)}$$

```
> F6:=convert(f6,polynom,x);
> F12:=convert(f12,polynom,x);
```

$$F6 := 2\,x - 2\,x^2 + 3\,x^3 - \frac{17}{6}\,x^4 + \frac{167}{60}\,x^5 - \frac{9}{4}\,x^6$$

$$F12 := 2\,x - 2\,x^2 + 3\,x^3 - \frac{17}{6}\,x^4 + \frac{167}{60}\,x^5 - \frac{9}{4}\,x^6 + \frac{637}{360}\,x^7 - \frac{12559}{10080}\,x^8 + \frac{5659}{6720}\,x^9 - \frac{479327}{907200}\,x^{10} + \frac{1269661}{3991680}\,x^{11} - \frac{2409641}{13305600}\,x^{12} \quad \textbf{(8)}$$

```
> g:=dsolve(IVP,y(x));  #This will give Maple's default solution
> G:=rhs(g);
> plot([G,F6,F12],x=-1..1,y=-2..2,numpoints=150,color=[green, red,
  blue]);
```

$$g := y(x) = -\left(4\,\mathrm{KummerU}\left(\frac{1}{3}, \frac{1}{2}, \frac{2}{3}\right)\mathrm{KummerM}\left(\frac{1}{3}, \frac{1}{2}, \frac{1}{6}\,(3\,x-2)^2\right)\right) \Big/ \left(\mathrm{KummerU}\left(\frac{1}{3}, \frac{1}{2}, \frac{2}{3}\right)\mathrm{KummerM}\left(-\frac{2}{3}, \frac{1}{2}, \frac{2}{3}\right) + 6\,\mathrm{KummerU}\left(-\frac{2}{3}, \frac{1}{2}, \frac{2}{3}\right)\mathrm{KummerM}\left(\frac{1}{3}, \frac{1}{2}, \frac{2}{3}\right)\right) + \left(4\,\mathrm{KummerM}\left(\frac{1}{3}, \frac{1}{2}, \frac{2}{3}\right)\mathrm{KummerU}\left(\frac{1}{3}, \frac{1}{2}, \frac{1}{6}\,(3\,x-2)^2\right)\right) \Big/ \left(\mathrm{KummerU}\left(\frac{1}{3}, \frac{1}{2}, \frac{2}{3}\right)\mathrm{KummerM}\left(-\frac{2}{3}, \frac{1}{2}, \frac{2}{3}\right) + 6\,\mathrm{KummerU}\left(\right.\right.$$

$$-\frac{2}{3}, \frac{1}{2}, \frac{2}{3}\Big) \operatorname{KummerM}\Big(\frac{1}{3}, \frac{1}{2}, \frac{2}{3}\Big)\Big)$$

$$G := -\Big(4 \operatorname{KummerU}\Big(\frac{1}{3}, \frac{1}{2}, \frac{2}{3}\Big) \operatorname{KummerM}\Big(\frac{1}{3}, \frac{1}{2}, \frac{1}{6}(3x-2)^2\Big)\Big) \Big/ \Big(\operatorname{KummerU}\Big(\frac{1}{3}, \frac{1}{2}, \frac{2}{3}\Big) \operatorname{KummerM}\Big(-\frac{2}{3}, \frac{1}{2}, \frac{2}{3}\Big) + 6 \operatorname{KummerU}\Big(-\frac{2}{3}, \frac{1}{2}, \frac{2}{3}\Big) \operatorname{KummerM}\Big(\frac{1}{3}, \frac{1}{2}, \frac{2}{3}\Big)\Big) + \Big(4 \operatorname{KummerM}\Big(\frac{1}{3}, \frac{1}{2}, \frac{2}{3}\Big) \operatorname{KummerU}\Big(\frac{1}{3}, \frac{1}{2}, \frac{1}{6}(3x-2)^2\Big)\Big) \Big/ \Big(\operatorname{KummerU}\Big(\frac{1}{3}, \frac{1}{2}, \frac{2}{3}\Big) \operatorname{KummerM}\Big(-\frac{2}{3}, \frac{1}{2}, \frac{2}{3}\Big) + 6 \operatorname{KummerU}\Big(-\frac{2}{3}, \frac{1}{2}, \frac{2}{3}\Big) \operatorname{KummerM}\Big(\frac{1}{3}, \frac{1}{2}, \frac{2}{3}\Big)\Big)$$



Example 3: Use dsolve and the series option.

```
> unassign('y');
> Order:=12;
```

$$Order := 12 \qquad (9)$$

```
> deq:=diff(y(x), x$2) + x*diff(y(x), x) + 2*y(x) = sin(x);
> inits:=y(1)=a[0],D(y)(1)=a[1];
> IVP:={deq,inits};
```

$$deq := \frac{d^2}{dx^2}\, y(x) + x\left(\frac{d}{dx}\, y(x)\right) + 2\, y(x) = \sin(x)$$

$$inits := y(1) = a_0, D(y)(1) = a_1$$

$$IVP := \left\{\frac{d^2}{dx^2}\, y(x) + x\left(\frac{d}{dx}\, y(x)\right) + 2\, y(x) = \sin(x), y(1) = a_0, D(y)(1) = a_1\right\} \quad \textbf{(10)}$$

```
> F:=dsolve(IVP,y(x),'series');
```

$$F := y(x) = a_0 + a_1\,(x-1) + \left(-\frac{1}{2}\,a_1 - a_0 + \frac{1}{2}\,\sin(1)\right)(x-1)^2 + \left(\frac{1}{3}\,a_0 - \frac{1}{6}\,\sin(1) - \frac{1}{3}\,a_1 + \frac{1}{6}\,\cos(1)\right)(x-1)^3 + \left(\frac{1}{4}\,a_1 + \frac{1}{4}\,a_0 - \frac{1}{6}\,\sin(1) - \frac{1}{24}\,\cos(1)\right)(x-1)^4 + \left(-\frac{2}{15}\,a_0 + \frac{3}{40}\,\sin(1) - \frac{1}{24}\,\cos(1) + \frac{1}{30}\,a_1\right)(x-1)^5 + \left(-\frac{1}{18}\,a_1 - \frac{1}{36}\,a_0 + \frac{1}{45}\,\sin(1) + \frac{11}{720}\,\cos(1)\right)(x-1)^6 + \left(\frac{11}{420}\,a_0 - \frac{79}{5040}\,\sin(1) + \frac{5}{1008}\,\cos(1) + \frac{1}{420}\,a_1\right)(x-1)^7 + \left(-\frac{5}{4032}\,\sin(1) + \frac{11}{1440}\,a_1 + \frac{1}{1440}\,a_0 - \frac{113}{40320}\,\cos(1)\right)(x-1)^8 + \left(-\frac{19}{5670}\,a_0 + \frac{761}{362880}\,\sin(1) - \frac{113}{362880}\,\cos(1) - \frac{13}{11340}\,a_1\right)(x-1)^9 + \left(-\frac{13}{181440}\,\sin(1) - \frac{37}{50400}\,a_1 + \frac{13}{50400}\,a_0 + \frac{1243}{3628800}\,\cos(1)\right)(x-1)^{10} + \left(\frac{311}{997920}\,a_0 - \frac{8111}{39916800}\,\sin(1) + \frac{181}{997920}\,a_1 + \frac{1}{39916800}\,\cos(1)\right)(x-1)^{11} + O\left((x-1)^{12}\right) \quad \textbf{(11)}$$

```
> # Here is how you convert this into a regular polynomial
> # (useful for plotting, if we had numbers in the initial
> # conditions).
> convert(rhs(F),polynom,x);
```

$$a_0 + a_1\,(x-1) + \left(-\frac{1}{2}\,a_1 - a_0 + \frac{1}{2}\,\sin(1)\right)(x-1)^2 + \left(\frac{1}{3}\,a_0 - \frac{1}{6}\,\sin(1) - \frac{1}{3}\,a_1 + \frac{1}{6}\,\cos(1)\right)(x-1)^3 + \left(\frac{1}{4}\,a_1 + \frac{1}{4}\,a_0 - \frac{1}{6}\,\sin(1) - \frac{1}{24}\,\cos(1)\right)(x-1)^4 + \left(-\frac{2}{15}\,a_0 + \frac{3}{40}\,\sin(1) - \frac{1}{24}\,\cos(1) + \frac{1}{30}\,a_1\right)(x-1)^5 + \left(-\frac{1}{18}\,a_1 - \frac{1}{36}\,a_0 + \frac{1}{45}\,\sin(1) + \frac{11}{720}\,\cos(1)\right)(x-1)^6 + \left(\frac{11}{420}\,a_0 - \frac{79}{5040}\,\sin(1) + \frac{5}{1008}\,\cos(1) + \frac{1}{420}\,a_1\right)(x-1)^7 + \left(-\frac{5}{4032}\,\sin(1) + \frac{11}{1440}\,a_1 + \frac{1}{1440}\,a_0\right. \quad \textbf{(12)}$$

$$- \frac{113}{40320} \cos(1) \Big) (x-1)^8 + \Big(- \frac{19}{5670} a_0 + \frac{761}{362880} \sin(1) - \frac{113}{362880} \cos(1)$$
$$- \frac{13}{11340} a_1 \Big) (x-1)^9 + \Big(- \frac{13}{181440} \sin(1) - \frac{37}{50400} a_1 + \frac{13}{50400} a_0$$
$$+ \frac{1243}{3628800} \cos(1) \Big) (x-1)^{10} + \Big(\frac{311}{997920} a_0 - \frac{8111}{39916800} \sin(1) + \frac{181}{997920} a_1$$
$$+ \frac{1}{39916800} \cos(1) \Big) (x-1)^{11}$$

Example 4: Construct a picture like Figure 5.2.4, p. 245, in Boyce and Diprima's book:

```
> unassign('y');

> deq:=diff(y(x),x$2)-x*y(x)=0;
> inits:=y(0)=0,D(y)(0)=1;
> IVP:={deq,inits};
```

$$deq := \frac{d^2}{dx^2} y(x) - x\, y(x) = 0$$
$$inits := y(0) = 0, D(y)(0) = 1$$
$$IVP := \left\{ \frac{d^2}{dx^2} y(x) - x\, y(x) = 0, y(0) = 0, D(y)(0) = 1 \right\} \quad \textbf{(13)}$$

```
> Order:=5;  #Make this one bigger than the poly degree
> f4:=dsolve(IVP,y(x),'series');
> F4:=convert(rhs(f4),polynom,x);
```

$$Order := 5$$
$$f4 := y(x) = x + \frac{1}{12} x^4 + O(x^5)$$
$$F4 := x + \frac{1}{12} x^4 \quad \textbf{(14)}$$

```
> #Now repeat those last two lines, changing the order
> # to create F4, F10, F16, F22

> Order:=11;
> f10:=dsolve(IVP,y(x),'series');
> F10:=convert(rhs(f10),polynom,x);
> Order:=17; f16:=dsolve(IVP,y(x),'series');
> F16:=convert(rhs(f16),polynom,x);
> Order:=23; f22:=dsolve(IVP,y(x),'series');
> F22:=convert(rhs(f22),polynom,x);
```

$$Order := 11$$
$$f10 := y(x) = x + \frac{1}{12} x^4 + \frac{1}{504} x^7 + \frac{1}{45360} x^{10} + O(x^{11})$$
$$F10 := x + \frac{1}{12} x^4 + \frac{1}{504} x^7 + \frac{1}{45360} x^{10}$$
$$Order := 17$$

$$f16 := y(x) = x + \frac{1}{12}x^4 + \frac{1}{504}x^7 + \frac{1}{45360}x^{10} + \frac{1}{7076160}x^{13} + \frac{1}{1698278400}x^{16} + \mathrm{O}(x^{17})$$

$$F16 := x + \frac{1}{12}x^4 + \frac{1}{504}x^7 + \frac{1}{45360}x^{10} + \frac{1}{7076160}x^{13} + \frac{1}{1698278400}x^{16}$$

$$Order := 23$$

$$f22 := y(x) = x + \frac{1}{12}x^4 + \frac{1}{504}x^7 + \frac{1}{45360}x^{10} + \frac{1}{7076160}x^{13} + \frac{1}{1698278400}x^{16} + \frac{1}{580811212800}x^{19} + \frac{1}{268334780313600}x^{22} + \mathrm{O}(x^{23})$$

$$F22 := x + \frac{1}{12}x^4 + \frac{1}{504}x^7 + \frac{1}{45360}x^{10} + \frac{1}{7076160}x^{13} + \frac{1}{1698278400}x^{16} + \frac{1}{580811212800}x^{19} + \frac{1}{268334780313600}x^{22} \qquad (15)$$

```
> F:=rhs(dsolve(IVP,y(x)));  #This is Maple's built-in solution
> #And plot them:
> plot({F4,F10,F16,F22,F},x=-10..2,y=-3..3);
```

$$F := -\frac{1}{3}\frac{\pi\, 3^{5/6}\,\mathrm{AiryAi}(x)}{\Gamma\left(\frac{2}{3}\right)} + \frac{1}{3}\frac{3^{1/3}\,\pi\,\mathrm{AiryBi}(x)}{\Gamma\left(\frac{2}{3}\right)}$$



3
2
y
1
-1
-2
-3
-10
-8
-6
-4
-2
0
2
x